

The program below resolves a problem with memory allocation that existed in version 2.0 EXOS and BASIC.

```
10 PROGRAM "Allocate_header"
20 !
30 ! For version 2.0 of BASIC,
40 ! this header will overcome
50 ! ALLOCATE bug. Lines 100,
60 ! 110, and 230 must be set
70 ! to suit the program.
80 !
100 LET UNIQUES="Invent your own meaningless string"
110 LET DESIRED_SPACE=200 ! For example
120 LET SIZE=DESIRED_SPACE+LEN(UNIQUES)
130 IF VERNUM=2 THEN
140   LET T=PEEK(544)+256*PEEK(545)-SIZE
150   ! Now T = Current allocation base - SIZE
160   FOR X=1 TO LEN(UNIQUES)
170     IF PEEK(X+T-1) <> ORD(UNIQUES(X:X)) THEN
180       ! This simply compares special header string to see if
program has already been run
190       ALLOCATE SIZE
200       POKE 542,PEEK(544)
210       POKE 543,PEEK(545)
220       ! Now the extension area base has been moved to equal
the current allocation base, to prevent automatic de-allocate
230       CODE="Invent your own meaningless string" ! This
string must be given here as a constant, identical to UNIQUES
240       RUN ! Re-runs program, but next time it won't re-
allocate, so that functions within program work properly
250       END IF
260     NEXT
270     LET X=X+T
280     ! Now X = Correct location counter
290     POKE 540,X BAND 255
300     POKE 541,X/256
310     ! Now BASIC's location counter is set correctly
320 ELSE ! When working properly
330   ALLOCATE DESIRED_SPACE
340 END IF
350 !
360 ! Remainder is simply test of operation
370 !
380 CODE FRED="real garbage"
390 PRINT FRED
400 DEF P(X)=PEEK(X)+256*PEEK(X+1)
410 FOR X=538 TO 544
420   PRINT X,P(X)
430 NEXT
```